/*=========================================================================
The Impact of Denying a Wanted Abortion on Women and Children
Authors: Juliana Londoño-Vélez and Estefanía Saravia

Creation date: January 9th, 2024
--------------------------------------------------------------------------
Figure 3 - Impacts Across Time
=========================================================================*/


use "$data/Final_Medellin_SisbenIV_Event_Study.dta", replace

label var Media "High school diploma"
label var Numero_Hijos "Number of children"
label var Familias_Accion "Familias en Acción recipient"
label var Enfermo_30 "Had a health problem (last 30 days)"
label var Grupo_AB "Extreme or moderate poverty"
label var Ocupado "Employed or looking for job"

rename (Enfermo_30 Numero_Hijos Media Ocupado Grupo_AB Familias_Accion)(PanelA PanelB PanelC PanelD PanelE PanelF)
global outcomes PanelA PanelB PanelC PanelD PanelE PanelF


gen FE_Edad = 1 if (Edad_Encuesta<18)
replace FE_Edad = 2 if (Edad_Encuesta>=18 & Edad_Encuesta<25)
replace FE_Edad = 3 if (Edad_Encuesta>=25 & Edad_Encuesta<35)
replace FE_Edad = 4 if (Edad_Encuesta>=35 & Edad_Encuesta<45)
replace FE_Edad = 5 if (Edad_Encuesta>=45)

gen Edad_squared=Edad_Encuesta^2

gen dis = 1 if distancia_encuesta<0
replace dis = 2 if distancia_encuesta<=4 & distancia_encuesta>=0
replace dis = 3 if distancia_encuesta>4 & distancia_encuesta<=8
replace dis = 4 if distancia_encuesta>8 

foreach var in $outcomes{
mat a=J(8,6,.)


local i=1
forvalues j=1/4{

ivreghdfe `var' Edad_Encuesta Edad_squared  (Niega=Mujer_Ponente)  if dis==`j', abs(oficina#time) cluster(id_juez) 

mat a[`i',1]=_b[Niega]
mat a[`i',2]=_se[Niega]
mat a[`i',3]=_b[Niega]-(_se[Niega])*1.96
mat a[`i',4]=_b[Niega]+(_se[Niega])*1.96
mat a[`i',5]=`i'
local i=`j'+1

}

preserve
svmat a

local title = proper("`var'")
local ylab: variable label `var'


	twoway (rcapsym  a4 a3 a5 , sort  lcolor(navy) mcolor(navy) msize(0.4) msymbol(+) lwidth(0.35)) (scatter a1 a5 , sort mcolor(black) msize(0.3) msymbol(circle) msize(medlarge)),  xlabel(1 "[-6,0)" 2 "[0,4)" 3 "[4,8)" 4 "[8,15]") xline(1.5, lpattern(dash)) yline(0, lcolor(gray)) legend(off) ylabel(,nogrid ) ytitle("`ylab'" "IV Coefficient and 95% CI") xtitle("Years since abortion denial") graphregion(fcolor(white) style(none) color(white) ///
 margin(r=6 l=3 t=3 b=5)) plotregion(fcolor(white) margin(r=7)) xsize(8)
 
	graph export "$output/Figure3_`var'.pdf", replace

	restore
	}
